

Esame di Stato Istituto Tecnico Industriale

Proposta di soluzione della parte facoltativa della seconda prova

Indirizzi: Informatica - CORSO DI ORDINAMENTO e Progetto "ABACUS"
Anno Scolastico: 2004-2005

La soluzione può essere organizzata mediante una serie di quattro pagine ASP, richiamate una dopo l'altra per eseguire le operazioni del testo con il seguente ordine:

generi.asp → *sottogeneri.asp* → *nomi_artisti.asp* → *album.asp*

Le pagine precedenti sono memorizzate nella directory *infoautori* pubblicata nel sito Web della casa discografica (cartella Web *casadiscografica*).

La base di dati (ad esempio il file di Access *CasaDiscografica.mdb*) è memorizzata nella cartella *fpdb* del sito Web; la **connessione al database** può essere dichiarata come variabile globale in modo che sia visibile durante una sessione HTTP di un utente all'applicazione Web. Per questo obiettivo è sufficiente creare un file di testo di nome *global.asa*, nella radice del sito Web, e dichiarare la variabile *StringaConnessione* nel corpo dell'evento *Session_OnStart* con la seguente sintassi.

```
global.asa
<SCRIPT LANGUAGE="VBScript" RUNAT="SERVER">
Sub Session_OnStart()
  Dim StringaConnessione
  Session("StringaConnessione") = "DRIVER={Microsoft Access Driver (*.mdb)};" & _
    "DBQ=" & Server.MapPath("fpdb/CasaDiscografica.mdb")
End Sub
</SCRIPT>
```

Per il richiamo automatico della pagina *generi.asp* possiamo inserire nella home folder del sito Web la seguente home page esemplificativa (utile per il collaudo delle operazioni realizzate). La pagina *default.asp*, oltre all'avvio delle operazioni richieste, ha anche la funzione di definire (tramite *global.asa*) la stringa di connessione al database.

```
default.asp
<%@ Language=VBScript %>
<% Response.Redirect("infoautori/generi.asp") %>
```

Le operazioni richieste dal testo possono essere realizzate, nell'ordine, nel seguente modo.

1. Una pagina con l'elenco dei generi musicali.

La prima pagina *generi.asp* nella directory Web *infoautori* esegue la seguente query presentando in una casella di riepilogo (creata dinamicamente nello script della pagina) la lista dei generi musicali disponibili.

```
SELECT DISTINCT Denominazione
FROM Generi
```

La pagina può essere organizzata nel seguente modo.



La casella di riepilogo è inserita in un form che richiama, quando l'utente fa clic sul pulsante, la pagina *sottogeneri.asp*. Il codice sorgente della pagina ASP *generi.asp* è il seguente.

```

infoautori/generi.asp
<%@ Language=VBScript %>
<% Response.Buffer=True %>

<HTML>
<HEAD><TITLE>Elenco dei generi</TITLE></HEAD>
<BODY>
<H3>Elenco dei generi</H3>
<% Dim ConnessioneDB, rsTabella, Query
   Set ConnessioneDB = Server.CreateObject("ADODB.Connection")
   ConnessioneDB.Open Session("StringaConnessione")

   Query = "SELECT DISTINCT Denominazione " & _
           " FROM Generi "

   Set rsTabella = ConnessioneDB.Execute(Query)%>
<% If rsTabella.BOF=True And rsTabella.EOF=True Then %>
  <H4>Nessun genere disponibile</H4>
<% Else %>
  <% Dim riga, denominazione %>
  <FORM ACTION="sottogeneri.asp" METHOD="POST">
    <SELECT MULTIPLE NAME="lstGeneri">
      <% Do until rsTabella.EOF=True
         denominazione = rsTabella.Fields("Denominazione")
         riga = "<OPTION VALUE='" & rsTabella.Fields("Denominazione") & "'> " & _
               denominazione & "</OPTION>"
         Response.Write riga
         rsTabella.MoveNext
       Loop
      %>
    </SELECT>
    <BR><INPUT TYPE="submit" NAME="cmdInvia" VALUE="Elenco sottogeneri">
  </FORM>
<% End If %>

<% rsTabella.Close : Set rsTabella = Nothing
   ConnessioneDB.Close : Set ConnessioneDB = Nothing
%>
</BODY>
</HTML>

```

2. Con un clic sul nome di un genere musicale, una pagina con l'elenco dei nomi dei sottogeneri musicali.

La seconda pagina *sottogeneri.asp* esegue la seguente query presentando in una casella di riepilogo (creata dinamicamente nello script della pagina) la lista dei sottogeneri musicali che corrispondono al genere scelto nella casella di riepilogo di *lstGeneri* inserita nella pagina *generi.asp*.

```

SELECT *
FROM Generi
WHERE Denominazione= [Genere]

```


Nella pagina *sottogeneri.asp*, il genere è ottenuto mediante il parametro *Genere*, il cui valore deriva dalla scelta (con un clic) dell'utente nella casella di riepilogo *lstGeneri* inserita nella pagina *generi.asp* (che richiama appunto *sottogeneri.asp*).

```

<% Genere = Request.Form("lstGeneri") %>

```

La pagina *sottogeneri.asp* può essere organizzata nel seguente modo.



La casella di riepilogo è inserita in un form che richiama, quando l'utente fa clic sul pulsante, la pagina *nomi_artisti.asp*. Il codice sorgente della pagina ASP *sottogeneri.asp* è il seguente.

```

infoautori/sottogeneri.asp
<%@ Language=VBScript %>
<% Response.Buffer=True %>

<HTML>
<HEAD><TITLE> Elenco sottogeneri </TITLE></HEAD>
<BODY>
<% Dim ConnessioneDB, rsTabella, Query, Genere
Set ConnessioneDB = Server.CreateObject("ADODB.Connection")
ConnessioneDB.Open Session("StringaConnessione")

Genere = Request.Form("lstGeneri") %>
<H3>Elenco dei sottogeneri del genere <%=Genere%></H3>

<% Query="SELECT * FROM Generi " &_
" WHERE Denominazione= '" & Genere & "'" ;"
Set rsTabella = ConnessioneDB.Execute(Query)%>
<% If rsTabella.BOF=True And rsTabella.EOF=True Then %>
<H4>Nessun sottogenere esistente </H4>
<% Else %>
<% Dim riga %>
<FORM ACTION="nomi_artisti.asp" METHOD="POST">
<SELECT MULTIPLE NAME="lstSottoGeneri">
<% Do until rsTabella.EOF=True
denominazione = rsTabella.Fields("SottoGenere")
riga = "<OPTION VALUE='" & rsTabella.Fields("SottoGenere") & "'> " &_
denominazione & "</OPTION>"
Response.Write riga
rsTabella.MoveNext
Loop
%>
</SELECT>
<BR><INPUT TYPE="submit" NAME="cmdInvia" VALUE="Elenco nomi artisti">
</FORM>
<% End If %>

<% rsTabella.Close : Set rsTabella = Nothing
ConnessioneDB.Close : Set ConnessioneDB = Nothing
%>
<P><A HREF="generi.asp">Home page</A></P>
</BODY>
</HTML>

```

3. Con un clic sul nome di un sottogenere musicale, una pagina con l'elenco dei nomi degli artisti.

La terza pagina *nomi_artisti.asp* esegue la seguente query presentando in una casella di riepilogo (creata dinamicamente nello script della pagina) la lista degli artisti che hanno scritto album appartenenti al sottogenere scelto nella casella di riepilogo di *lstSottoGeneri* inserita nella pagina *sottogeneri.asp*.

```
SELECT DISTINCT Artisti.Denominazione
FROM Generi INNER JOIN
    (Artisti INNER JOIN Album ON Artisti.IDArtista = Album.IDArtista)
    ON Generi.IDGenere = Album.IDGenere
WHERE Generi.SottoGenere= [SottoGenere]
```

Nella pagina *nomi_artisti.asp*, il sottogenere è ottenuto mediante il parametro *SottoGenere*, il cui valore deriva dalla scelta (con un clic) dell'utente nella casella di riepilogo *lstSottoGeneri* inserita nella pagina *sottogeneri.asp* (che richiama appunto *nomi_artisti.asp*).

```
<% SottoGenere = Request.Form("lstSottoGeneri") %>
```

La pagina *nomi_artisti.asp* può essere organizzata nel seguente modo.



La casella di riepilogo è inserita in un form che richiama, quando l'utente fa clic sul pulsante, la pagina *album.asp*. Il codice sorgente della pagina ASP *nomi_artisti.asp* è il seguente.

```
infoautori/nomi_artisti.asp
<%@ Language=VBScript %>
<% Response.Buffer=True %>

<HTML>
<HEAD><TITLE>Elenco nomi artisti</TITLE></HEAD>
<BODY>
<% Dim ConnessioneDB, rsTabella, Query, SottoGenere
Set ConnessioneDB = Server.CreateObject("ADODB.Connection")
ConnessioneDB.Open Session("StringaConnessione")
SottoGenere = Request.Form("lstSottoGeneri") %>

<H3>Elenco dei nomi degli artisti del sottogenere <%=SottoGnere%> </H3>

<% Query = "SELECT DISTINCT Artisti.Denominazione " &_
" FROM Generi INNER JOIN " &_
" (Artisti INNER JOIN Album ON Artisti.IDArtista=Album.IDArtista) " &_
" ON Generi.IDGenere=Album.IDGenere " &_
" WHERE Generi.SottoGenere='" & SottoGenere & "';"

Set rsTabella = ConnessioneDB.Execute(Query)%>
<% If rsTabella.BOF=True And rsTabella.EOF=True Then %>
<H4>Nessun artista disponibile</H4>
<% Else %>
<% Dim riga, denominazione %>
```

```

<FORM ACTION="album.asp" METHOD="POST">
  <SELECT MULTIPLE NAME="lstArtisti">
    <% Do until rsTabella.EOF=True
      denominazione = rsTabella.Fields("Denominazione")
      riga = "<OPTION VALUE='" & rsTabella.Fields("Denominazione") & "'> " &_
        denominazione & "</OPTION>"
      Response.Write riga
      rsTabella.MoveNext
    Loop %>
  </SELECT>
  <BR><INPUT TYPE="submit" NAME="cmdInvia" VALUE="Elenco album">
</FORM>
<% End If %>

<% rsTabella.Close : Set rsTabella = Nothing
  ConnessioneDB.Close : Set ConnessioneDB = Nothing
%>
<P><A HREF="generi.asp">Home page</A></P>
</BODY>
</HTML>

```

4. Con un clic sul nome di un artista, l'elenco degli album di quell'artista con le immagini di copertina, la durata, l'anno o gli anni di edizione, le notizie bibliografiche sull'artista ed il periodo di attività.

La quarta pagina *album.asp* esegue la seguente query presentando una tabella con le informazioni richieste relative all'artista scelto nella casella di riepilogo di *lstArtisti* inserita nella pagina *nomi_artisti.asp*.

```

SELECT Album.TitoloAlbum, Edizioni.Copertina, Album.Durata,
  Edizioni.Data, Artisti.NotizieBibliografiche, Artisti.PeriodoAttività
FROM Artisti INNER JOIN
  (Album INNER JOIN Edizioni ON Album.IDAlbum = Edizioni.IDAlbum)
  ON Artisti.IDArtista = Album.IDArtista
WHERE Artisti.Denominazione= [NomeArtista]

```

Nella pagina *album.asp*, il nome dell'artista è ottenuto mediante il parametro *NomeArtista*, il cui valore deriva dalla scelta (con un clic) dell'utente nella casella di riepilogo *lstArtisti* inserita nella pagina *nomi_artisti.asp* (che richiama appunto *album.asp*).

```

<% NomeArtista = Request.Form("lstArtisti") %>

```

La pagina *album.asp* presenta le informazioni richieste in una tabella organizzata nel seguente modo.

Indirizzo <http://localhost/casadiscografica/infoautori/album.asp> Vai Collegamenti >>

Elenco degli album dell'artista GRUPPO A

Titolo album	Copertina	Durata album	Data edizione	Notizie bibliografiche artista	Peridodo attività
Titolo album A	Copertina A	60 m 25 s	02/02/2002	Notizie GRUPPO A	10
Titolo album A	Copertina B	60 m 25 s	03/04/2004	Notizie GRUPPO A	10
Titolo album B	Copertina C	56 m 4 s	06/07/2004	Notizie GRUPPO A	10

[Home page](#)

Il codice sorgente della pagina ASP *album.asp* è il seguente.

```

infoautori/album.asp
<%@ Language=VBScript %>
<% Response.Buffer=True %>

<HTML>
<HEAD><TITLE>Album di un artista</TITLE></HEAD>
<BODY>
<% Dim ConnessioneDB, rsTabella, Query, NomeArtista
   Set ConnessioneDB = Server.CreateObject("ADODB.Connection")
   ConnessioneDB.Open Session("StringaConnessione")
   NomeArtista = Request.Form("lstArtisti") %>

   <H3>Elenco degli album dell'artista <%=NomeArtista%></H3>

   <% Query = "SELECT Album.TitoloAlbum, Edizioni.Copertina, Album.Durata, " &_
      " Edizioni.Data, Artisti.NotizieBibliografiche, Artisti.PeriodoAttività " &_
      " FROM Artisti INNER JOIN " &_
      " (Album INNER JOIN Edizioni ON Album.IDAlbum = Edizioni.IDAlbum) " &_
      " ON Artisti.IDArtista = Album.IDArtista " &_
      " WHERE Artisti.Denominazione=' " & NomeArtista & "';"

   Set rsTabella = ConnessioneDB.Execute(Query)%>
   <% If rsTabella.EOF=True And rsTabella.BOF=True Then %>
     <H4>Nessun album disponibile</H4>
   <% Else %>
     <TABLE BORDER="1">
       <TR>
         <TH>Titolo album</TH><TH>Copertina</TH><TH>Durata album</TH> <TH>Data edizione</TH>
         <TH>Notizie bibliografiche artista</TH><TH>Peridodo attività</TH>
       </TR>
       <% Dim riga, URLcopertina
          Do until rsTabella.EOF=True %>
         <TR>
           <TD ALIGN="CENTER"><%=rsTabella.Fields("TitoloAlbum")%></TD>
           <% URLcopertina = "images/" & rsTabella.Fields("copertina")
              riga = "<TD ALIGN='CENTER'><IMG SRC=" & URLcopertina & "></TD>"
              Response.Write riga
           %>
           <TD ALIGN="CENTER"><%=rsTabella.Fields("Durata")%></TD>
           <TD ALIGN="CENTER"><%=rsTabella.Fields("Data")%></TD>
           <TD ALIGN="CENTER"><%=rsTabella.Fields("NotizieBibliografiche")%></TD>
           <TD ALIGN="CENTER"><%=rsTabella.Fields("PeriodoAttività")%></TD>
         </TR>
         <% rsTabella.MoveNext %>
       <% Loop %>
     </TABLE>
   <% End If %>

   <% rsTabella.Close : Set rsTabella = Nothing
      ConnessioneDB.Close : Set ConnessioneDB = Nothing
   %>
   <P><A HREF="generi.asp">Home page</A></P>
</BODY>
</HTML>

```